



isel-Zwischenformat

Befehlsübersicht

isel <--> *automation*

Software-Handbuch

B.ZF.11.98

Übersicht des Formats zur Ansteuerung von *isel*-Maschinen

NCP-Dateikennung	IMF_PBLxxxxxxxxxx Angabe einer Kennung für eine gültige NCP-Quelldatei. Die Angabe dieser Kennung ist nur erforderlich, falls mit einer Servo-Steuerung Bahnbewegungen realisiert werden sollen. Die Kennung steht am Anfang der ersten Zeile der NCP-Quelldatei.
Groß-/ Kleinschreibung	beliebig keine Unterscheidung zwischen Groß- und Kleinschreibung.
Achsbezeichner	X Y Z A B C Zulässige Achsbezeichner sind X, Y und Z sowie A für die 4. Achse, B für die 5. Achse usw.
Kommentar	; Einleitung eines Kommentars. Der Kommentar gilt bis zum Ende der aktuellen Zeile.
Trennzeichen	Zwischen den Kommandos und den einzelnen Parametern sind sogenannte „white spaces“ als Trennzeichen zulässig (das sind Formatierungszeichen wie Leerzeichen, TABs usw.).
Zeilennummer	Nxxxxxx Angabe einer Zeilennummer des aktuellen NC-Satzes. Für die spätere Verwendung des Bahndatengenerators einer Servo-Steuerung ist die Angabe von Zeilennummern in der NCP-Quelldatei notwendig. Der Bahndatengenerator einer Microstep-Steuerung kann Dateien mit und ohne Zeilennummern bearbeiten.
Referenzfahrt	REF X (Y Z A ...) Ausführen einer Referenzfahrt der angegebenen Achsen. Die Reihenfolge der Angabe ist beliebig. Die Referenzfahrt wird in der Reihenfolge Z Y X A B ... ausgeführt. Um die Reihenfolge zu ändern, fügen Sie ein Whitespace-Trennzeichen zwischen Achsbezeichnern ein. <u>Beispiel</u> <i>REF X ;Ausführen der Referenzfahrt X</i> <i>REF XYZ ;Referenzfahrt in der Reihenfolge Z Y X</i> <i>REF X Y Z ;Referenzfahrt in der Reihenfolge X Y Z</i> <i>REF X YZA ;Erst Referenzfahrt X, dann in der Reihenfolge Z Y A</i>
SW-Endschalter ^{*)}	LIMIT Xneg,pos Yneg,pos Zneg,pos Aneg,pos Definieren der Software-Endschalter für den nachfolgenden Verfahrbetrieb. Nach dem Achsbezeichner folgt der negative sowie der positive Wert für den Software-Endschalter. Die Angabe der Positionen erfolgt in Mikrometer [µm].

*) siehe Seite 14

Normal- geschwindigkeit	<p>VEL xxx Setzen der Normalgeschwindigkeit auf den Wert xxx. Die Normalgeschwindigkeit ist die Segment-Verfahrgeschwindigkeit, die bei Verwendung eines der Befehle MOVEABS, MOVEREL, FASTABS, FASTREL, CWABS, CWREL, CCWABS, CCWREL, CWHLXABS, CWHLXREL, CCWHLXABS oder CCWHLXREL benutzt wird. Die Einheit des übergebenen Geschwindigkeitswertes ist [$\mu\text{m/s}$].</p> <p><u>Beispiel</u> <i>VEL 5000 ;setzt die Normalgeschwindigkeit auf 5 mm/s</i></p>
Eilgeschwindigkeit	<p>FASTVEL xxx Setzen der Eilgeschwindigkeit auf den Wert xxx. Die Eilgeschwindigkeit ist die Segment-Verfahrgeschwindigkeit, die bei Verwendung eines der Befehle FASTABS oder FASTREL benutzt wird. Die Einheit des übergebenen Geschwindigkeitswertes ist [$\mu\text{m/s}$].</p> <p><u>Beispiel</u> <i>FASTVEL 50000 ;setzt die Eilgeschwindigkeit auf 50 mm/s</i></p>
Absolutbewegung	<p>MOVEABS X... Y... Z... A... Absolute Linearbewegung mit Normalgeschwindigkeit (das ist die Geschwindigkeit, die durch Verwendung des Befehls VEL gesetzt wird). Diese Funktion führt eine Bewegung von bis zu vier Achsen aus. Bei einer Konfiguration der Achsen als XYZ_TTT-Struktur werden die drei Achsen X, Y und Z interpoliert; die Achse A wird synchron nachgeführt. Durch den Achsbezeichner (X, Y, ...) wird die Zielcoordinate der entsprechenden Achse zugewiesen. Die Zielposition der Bewegung wird in Absolutkoordinaten angegeben (bezogen auf den aktuell gesetzten Werkstücknullpunkt). Die Angabe der Zielposition für jede Achse ist modal, d. h. wenn für eine oder mehrere Achsen kein Wert übergeben wird, wird die aktuelle Position beibehalten. Die Einheit der Zielposition ist Mikrometer [μm] für Linearachsen und Bogensekunden ['] für Drehachsen ($1^\circ = 60' = 3600''$).</p> <p><u>Beispiel</u> <i>VEL 10000 ;Geschw. 10 mm/s</i> <i>MOVEABS X20000 Y15000 Z-5000 ;Achsen XYZ an neue Position</i> <i>MOVEABS Y15000 ;nur Achse Y an neue Position</i> <i>;andere Achsen bleiben stehen</i></p>
Abs. Eilbewegung	<p>FASTABS X... Y... Z... A... Bewegung zu einer Absolutposition in Eilgeschwindigkeit (das ist die Geschwindigkeit, die durch Verwendung des Befehls FASTVEL gesetzt wird). Die Bedeutung der nachstehenden Parameter ist die gleiche wie bei der Absolutbewegung mit Normalgeschwindigkeit.</p>

- Relativbewegung** **MOVEREL X... Y... Z... A...**
 Relative Linearbewegung mit Normalgeschwindigkeit (das ist die Geschwindigkeit, die durch Verwendung des Befehls VEL gesetzt wird).
 Diese Funktion führt eine Bewegung von bis zu vier Achsen aus. Bei einer Konfiguration der Achsen als XYZ_TTT-Struktur werden die drei Achsen X, Y und Z interpoliert; die Achse A wird synchron nachgeführt. Durch den Achsbezeichner (X, Y, ...) wird die Zielkoordinate der entsprechenden Achse zugewiesen. Die Zielposition der Bewegung wird in Absolutkoordinaten angegeben (bezogen auf die aktuelle Position der Achsen).
 Die Angabe der Zielposition für jede Achse ist modal, d. h. wenn für eine oder mehrere Achsen kein Wert übergeben wird, wird für diese Achse(n) automatisch der Verfahrweg 0 angenommen.
 Die Einheit der Zielposition ist Mikrometer [µm] für Linearachsen und Bogensekunden ['] für Drehachsen (1° = 60' = 3600'').
- Beispiel
- | | |
|-------------------------------------|----------------------------------|
| <i>VEL 5000</i> | <i>;Geschwindigkeit 5 mm/s</i> |
| <i>MOVEREL X20000 Y15000 Z-5000</i> | <i>;Bewegung von 3 Achsen</i> |
| <i>MOVEREL Z-5000</i> | <i>;nur Achse Z 5 mm abwärts</i> |
-
- Rel. Eilbewegung** **FASTREL X... Y... Z... A...**
 Relative Bewegung in Eilgeschwindigkeit (das ist die Geschwindigkeit, die durch Verwendung des Befehls FASTVEL gesetzt wird)
 Die Bedeutung der nachstehenden Parameter ist die gleiche wie bei der Relativen Normalbewegung.
- Beispiel
- | | |
|--------------------------------|------------------------------------|
| <i>VEL 5000</i> | <i>;Normalgeschwind. 5 mm/s</i> |
| <i>FASTVEL 35000</i> | <i>;Eilgeschwindigkeit 35 mm/s</i> |
| <i>MOVEREL Z5000</i> | <i>;Achse Z 5 mm anheben</i> |
| <i>FASTREL X100000 Y100000</i> | <i>;Positionierung in XY</i> |
-
- Kreisebene** **PLANE XY (XZ, YZ)**
 Festlegen der Bearbeitungsebene für die Kreisinterpolation bei Verwendung eines der Befehle CWABS, CCWABS, CWREL oder CCWREL.
 Dieser Befehl gilt nur für die Bearbeitungsebene des Kreises, er kann nicht zur Festlegung einer Arbeitsebene der Linearbewegungen verwendet werden. Das Verhalten des Befehls ist modal, d. h. wenn die Kreisebene einmal gesetzt ist, bleibt diese Einstellung bis zum nächsten Aufruf des Befehls PLANE erhalten.
- Beispiel
- | | |
|------------------------------------|---|
| <i>PLANE XY</i> | <i>;für die Kreisinterpolation werden
;die Achsen X und Y benutzt</i> |
| <i>CWREL I20000 J0 X0 Y0 Z0 A0</i> | <i>;Kreisinterpolation ausführen</i> |
| <i>PLANE YZ</i> | <i>;die Achsen Y und Z werden benutzt</i> |
| <i>CWREL I20000 J0 X0 Y0 Z0 A0</i> | <i>;mit dem gleichen Kommando
;führen Y und Z den Kreis aus</i> |

Kreis Absolut

CWABS/CCWABS I... J... X... Y.... Z.... A....

Ausführen einer Kreisinterpolation Kreisinterpolation in einer der zulässigen Ebenen XY, YZ oder XZ unter Verwendung von Absolutkoordinaten.

Die Angabe der Koordinaten erfolgt absolut, d. h. alle Positionsangaben beziehen sich auf den gesetzten Werkstücknullpunkt.

Die Festlegung der Kreisrichtung erfolgt mit dem Kommando CWABS (im Uhrzeigersinn) oder CCWABS (im Gegen-Uhrzeigersinn).

Die Parameter I und J geben die Koordinaten des Kreismittelpunktes an; I und J sind Abhängigkeit von der jeweils aktiven Kreisebene.

Bearbeitungsebene	Bedeutung
XY-Ebene	I entspricht einer X-, J einer Y-Koordinate
XZ-Ebene	I entspricht einer X-, J einer Z-Koordinate
YZ-Ebene	I entspricht einer Y-, J einer Z-Koordinate

Die Parameter X, Y, Z, und A geben die jeweiligen absoluten Endpositionen an.

Falls eine Achse ihre aktuelle Position behalten soll, kann die Angabe der Endposition für diese Achse entfallen. Falls Angaben für Achsen gemacht werden, die für die Kreisinterpolation nicht benötigt werden, werden diese ignoriert.

Die Angabe der Positionen erfolgt in Mikrometer. Als Fahrgeschwindigkeit wird die Normalgeschwindigkeit verwendet (das ist die Geschwindigkeit, die durch Verwendung des Befehls VEL gesetzt wird).

Beispiel

```

PLANE XY ;Festlegen der Interpolationsebene
MOVEABS X50000 Y50000 Z-2500 ;Startpunkt anfahren
CWABS I75000 J50000 X50000 Y50000 Z-2500 ;Kreis mit r = 25 mm
MOVEABS X50000 Y50000 Z-2500 ;dieser Befehl erzeugt die
CWABS I75000 ;gleiche Bewegung, aber nur
;die absolut notwendigen
;Parameter werden übergeben
  
```

Kreis Relativ

CWREL/CCWREL I... J... X... Y.... Z.... A....

Dieser Befehl führt eine Kreisinterpolation unter Verwendung von Relativkoordinaten aus.

Die Verwendung der Kreisparameter erfolgt sinngemäß wie bei der Ausführung mit der Angabe von Absolutkoordinaten mit dem Unterschied, daß die Parameter relative Positionen beschreiben. Die Angabe der Parameter erfolgt modal, d. h. wenn eine relative Endposition 0 ist, kann diese Angabe entfallen. Falls Parameter X, Y, Z, A, B und C auftauchen, die nicht benötigt werden, werden diese ignoriert.

```

PLANE XY ;Festlegen der Interpolationsebene
MOVEABS X50000 Y50000 Z-2500 ;Startpunkt anfahren
CWREL I20000 J0 X0 Y0 Z0 A0 ;Vollkreis mit Radius 20 mm
MOVEABS X50000 Y50000 Z-2500 ;derselbe Kreis, aber ohne die
CWREL I20000 ;nicht erforderlichen Parameter
  
```

- Zeitverzögerung*) **WAIT xxx**
 Ausführen einer Zeitverzögerung während der Bearbeitung.
 Die Angabe der Zeit erfolgt in Millisekunden [ms].
 Während der Ausführung des Befehls kann die Zeitverzögerung je nach Art des verwendeten Steuerungsprogramms übersprungen bzw. abgebrochen werden.
Beispiel *WAIT 2000 ;2 Sekunden warten*
- WS-Null Setzen **WPZERO**
 Die Position, an dem das Werkzeug im Augenblick steht, wird als neuer Werkstücknullpunkt gesetzt. Der alte Werkstücknullpunkt wird gelöscht. Es werden keine Parameter benötigt.
Beispiel
REF XYZ ;Referenzfahrt
MOVEABS X30000 Y25000 ;An eine Absolutposition fahren
WPZERO ;und diese Position als neuen
;Werkstücknullpunkt setzen
- WS-Null Absolut **WPZEROABS X... Y... Z... A...**
 Die übergebenen Koordinaten werden als neuer Werkstücknullpunkt gesetzt. Die Koordinaten sind Absolutwerte und beziehen sich auf den Maschinennullpunkt (das ist die Position, die normalerweise durch eine Referenzfahrt ermittelt wird). Der alte Werkstücknullpunkt wird gelöscht. Die Angabe der Positionen erfolgt in Mikrometer [µm].
 Falls die Position des Werkstücknullpunktes für eine oder mehrere Achsen unverändert bleiben soll, genügt es, die entsprechenden Parameter auszulassen. Für diese Achse wird die bisher benutzte Koordinate beibehalten.
Beispiel
REF XYZ ;Referenzfahrt von XYZ
WPZEROABS X5000 Y5000 Z-5000 ;Werkstücknullpunkt an (X5000,
;Y5000, Z-5000) setzen
WPZEROABS X35000 Y20000 ;jetzt an Position (X35000,
;Y20000, Z-5000) setzen
- Clear WS-Nullpkt **WPCLEAR**
 Löschen des aktuell gesetzten Werkstücknullpunktes. Der Maschinennullpunkt (Position nach Referenzfahrt) wird als neuer Werkstücknullpunkt angenommen. Diese Funktion erfordert keine Parameter.
Beispiel
WPCLEAR ;Werkstücknullpunkt löschen

*) siehe Seite 14

Laden WS-Nullpkt*) **WPREGn X... Y... Z... A...**

Laden einer Werkstücknullpunktposition in das Werkstücknullpunktregister n (zum späteren Gebrauch). Der Parameter n ist der Index des benutzten Registers.

Die übergebenen Koordinaten sind Absolutwerte und beziehen sich auf den Maschinennullpunkt (Position nach Referenzfahrt).

Die einzelnen Werkstücknullpunkte können mit der Funktion WPREGnACT aktiviert werden, wobei 'n' den Index des gewünschten Registers angibt.

In Abhängigkeit vom benutzten Steuerungsprogramm können zwei oder mehr Werkstücknullpunkte verwaltet werden:

n = 2 falls Sie PRO-PAL oder PRO-DIN als Steuerungsprogramm benutzen

n = 8 falls Sie REMOTE für Servomotoranlagen oder REMOTE für Schrittmotoranlagen benutzen

Beispiel

```

WPREG1 X100000 Y50000 Z-50000 ;Register 1 mit Pos. laden
WPREG2 X100000 Y100000 Z-50000 ;Register 2 mit Pos. laden
WPREG1ACT ;Register 1 aktivieren
MOVEABS X0 Y0 Z0 ;an die Position (0,0,0) fahren
WPREG2ACT ;Register 2 aktivieren
MOVEABS X0 Y0 Z0 ;an die Position (0,0,0) fahren,
;die Position Y ist nun um 50 mm verschoben
  
```

 WS-Nullpkt-Reg
aktivieren*)

WPREGnACT

Diese Funktion aktiviert den Werkstücknullpunkt, der zuvor in das Register 'n' geladen wurde. Der Parameter 'n' ist der Index des gewünschten Registers, er ist im Unterschied zu den meisten anderen Befehlen Bestandteil des Kommandos. Es dürfen also keine Leerzeichen vor oder hinter n stehen.

Beispiel siehe Befehl WPREGn

 WS-Null teachen*) **WPTEACH**

Während der Bearbeitung wird eine Teachbox geöffnet, mit deren Hilfe Sie einen neuen Werkstücknullpunkt anteachen können. Mit OK wird die aktuelle Position als neuer Werkstücknullpunkt übernommen; das Beenden mit ESC behält den alten Werkstücknullpunkt bei.

Dieser Befehl steht nur zur Verfügung, wenn Sie das Programm REMOTE für Servomotoranlagen oder REMOTE für Schrittmotoranlagen benutzen.

Beispiel

```

WPCLEAR ;aktuellen Werkstücknullpunkt löschen
MOVEABS X0 Y0 Z-0 ;an dieser Stelle wird ein neuer
;Werkstücknullpunkt benötigt
WPTEACH ;Öffnen der Teachbox
MOVEABS X10000 Y20000 ;Weiter mit der Bearbeitung
MOVEABS Z-15000
  
```

Port setzen*)

SETPORT An=v

Setzen eines Ausgabeports.

Die Zuweisung einer physikalischen Adresse zu einer logischen Adresse wird im Einstellprogramm des Treibers vorgenommen. Das Einrichten der logischen Adresse ist für die Benutzung dieser Funktion erforderlich.

Der Parameter 'n' (Ziffer zwischen 1 und 4) gibt die logische Adresse eines Ausgabeports an. Der Parameter 'v' ist der auszugebende Wert. Das Format des Ausgabewertes richtet sich nach dem angehängten Buchstaben.

Beispiel

```
SETPORT A1=100101B      ;setzt den Binärwert 00100101
SETPORT A1=42D          ;setzt den Dezimalwert 42
SETPORT A1=F2H          ;setzt den Hexadezimalwert F2
```

Bit setzen*)

SETBIT An.b

Setzen eines einzelnen Bits eines Ausgabeports.

Die Zuweisung einer physikalischen Adresse zu einer logischen Adresse wird im Einstellprogramm des Treibers vorgenommen. Das Einrichten der logischen Adresse ist für die Benutzung dieser Funktion erforderlich.

Der Parameter 'n' (Ziffer zwischen 1 und 4) gibt die logische Adresse des Ausgabeports an, der Parameter 'b' (Ziffer zwischen 1 und 8) kennzeichnet die Bitnummer des zu setzenden Bits.

Beispiel

```
SETBIT A1.4             ;setzt Bit 4 von Ausgabeport 1 auf „1“
SETBIT A2.1             ;setzt Bit 1 von Ausgabeport 2 auf „1“
```

Bit löschen*)

RESBIT An.b

Löschen eines einzelnen Bits eines Ausgabeports.

Die Zuweisung einer physikalischen Adresse zu einer logischen Adresse wird im Einstellprogramm des Treibers vorgenommen. Das Einrichten der logischen Adresse ist für die Benutzung dieser Funktion erforderlich.

Der Parameter n (eine Ziffer zwischen 1 und 4) gibt die logische Adresse des Ausgabeports an, der Parameter b (eine Ziffer zwischen 1 und 8) kennzeichnet die Bitnummer des zu löschenden Bits.

Beispiel

```
RESBIT A1.4             ;setzt Bit 4 von Ausgabeport 1 auf „0“
RESBIT A2.1             ;setzt Bit 1 von Ausgabeport 2 auf „0“
```

*) siehe Seite 14

Warten auf Bitwert*) WAITBIT Ep.n==v

Warten auf einen Bitwert.

Die Zuweisung einer physikalischen Adresse zu einer logischen Adresse wird im Einstellprogramm des Treibers vorgenommen. Das Einrichten der logischen Adresse ist für die Benutzung dieser Funktion erforderlich.

Das Steuerungsprogramm stoppt die Bearbeitung und wartet, bis der definierte Bitwert am angegebenen Eingangskanal auftritt. Der Parameter 'E' gibt die logische Portadresse (Ziffer zwischen 1 und 4) sowie Bitnummer (Ziffer zwischen 1 und 8) des zu prüfenden Eingangskanals an. Der Parameter 'v' hinter dem Vergleichsoperator ist der logische Pegel, auf den gewartet werden soll.

Für Test- oder Kontrollzwecke ist es möglich, diese Funktion abzubrechen bzw. zu überspringen.

Beispiel

```
MOVEABS X0 Y0           ;Bearbeitung ...
WAITBIT E1.4==1         ;Warten bis Bit 4 an Port 1 high wird
MOVEABS X20000 Y5000    ;weiter mit der Bearbeitung ...
```

**Warten auf
Bitmuster*)****WAITPORT Ep==v**

Warten auf Bitmuster an einem Eingangsport

Die Zuweisung einer physikalischen Adresse zu einer logischen Adresse wird im Einstellprogramm des Treibers vorgenommen. Das Einrichten der logischen Adresse ist für die Benutzung dieser Funktion erforderlich.

Das Steuerungsprogramm stoppt die Bearbeitung und wartet, bis das definierte Bitmuster am angegebenen Eingangsport auftritt. Der Parameter 'E' gibt die logische Portadresse an (Ziffer zwischen 1 und 4). Der Parameter 'v' hinter dem Vergleichsoperator ist der 8-Bit-Wert, auf den gewartet werden soll.

Für Test- oder Kontrollzwecke ist es möglich, diese Funktion abzubrechen bzw. zu überspringen.

Beispiel

```
MOVEABS Z0               ;Bearbeitung ...
MOVEABS X0 Y0           ;Bearbeitung ...
WAITPORT E1==00100101B  ;warte auf Binärwert 00100101
oder
WAITPORT E1==37D        ;warte auf Dezimalwert 37
oder
WAITPORT E1==25D        ;warte auf Hexadezimalwert 25
MOVEABS X20000 Y5000    ;weiter mit der Bearbeitung ...
```

*) siehe Seite 14

<p>Beginn Bahnbearbeitung*)</p>	<p>PATH</p> <p>Startet die Bearbeitung eines Bahndatenfeldes. Das Starten und Beenden der Bearbeitung eines Bahndatenfeldes sind intern benutzte Funktionen für einen besonderen Arbeitsmodus des Steuerungsprogrammes.</p> <p>Ein Bahndatenfeld bzw. eine Datei mit Bahndatenfeldern wird durch ein spezielles Hilfsprogramm von <i>iselautomation</i> erstellt, einem sogenannten Bahndatengenerator. Anwender sollten auf keinen Fall eigene Bahndatenfelder generieren.</p>																		
<p>Ende Bahnbearbeitung*)</p>	<p>PATHEND</p> <p>Kennzeichnet das Ende eines Bahndatenfeldes. Das Starten und Beenden der Bearbeitung eines Bahndatenfeldes sind intern benutzte Funktionen für einen besonderen Arbeitsmodus des Steuerungsprogrammes.</p> <p>Ein Bahndatenfeld bzw. eine Datei mit Bahndatenfeldern wird durch ein spezielles Hilfsprogramm von <i>iselautomation</i>, einen sogenannten Bahndatengenerator, erstellt. Anwender sollten auf keinen Fall eigene Bahndatenfelder generieren.</p>																		
<p>Werkzeugwechsel*)</p>	<p>GETTOOL x</p> <p>Ruft das Hilfsprogramm für den automatischen Werkzeugwechsel auf. Der Parameter x gibt die Nummer des neuen Werkzeuges an. Bevor Sie diesen Befehl benutzen können, müssen Sie das Werkzeugwechselprogramm (ITC) selbst einrichten sowie die Benutzung des Werkzeugwechselprogramms in der Steuerungsoberfläche (REMOTE) festlegen.</p> <p><u>Beispiel</u></p> <table border="0"> <tr> <td><i>MOVEABS Z5000</i></td> <td><i>;Bearbeitung ...</i></td> </tr> <tr> <td><i>FASTABS X5000 Y5000</i></td> <td><i>;Bearbeitung ...</i></td> </tr> <tr> <td><i>COOLANT OFF</i></td> <td><i>;Kühlmittelpumpe aus</i></td> </tr> <tr> <td><i>SPINDLE OFF</i></td> <td><i>;Spindel aus</i></td> </tr> <tr> <td><i>GETTOOL 3</i></td> <td><i>;Neues Werkzeug 3 holen</i></td> </tr> <tr> <td><i>SPINDLE ON</i></td> <td><i>;Spindel wieder einschalten</i></td> </tr> <tr> <td><i>MOVEABS Z10000</i></td> <td><i>;Sicherheitshöhe</i></td> </tr> <tr> <td><i>FASTABS X150000 Y200000</i></td> <td><i>;Positionieren ...</i></td> </tr> <tr> <td><i>COOLANT ON</i></td> <td><i>;Kühlmittelpumpe ein</i></td> </tr> </table>	<i>MOVEABS Z5000</i>	<i>;Bearbeitung ...</i>	<i>FASTABS X5000 Y5000</i>	<i>;Bearbeitung ...</i>	<i>COOLANT OFF</i>	<i>;Kühlmittelpumpe aus</i>	<i>SPINDLE OFF</i>	<i>;Spindel aus</i>	<i>GETTOOL 3</i>	<i>;Neues Werkzeug 3 holen</i>	<i>SPINDLE ON</i>	<i>;Spindel wieder einschalten</i>	<i>MOVEABS Z10000</i>	<i>;Sicherheitshöhe</i>	<i>FASTABS X150000 Y200000</i>	<i>;Positionieren ...</i>	<i>COOLANT ON</i>	<i>;Kühlmittelpumpe ein</i>
<i>MOVEABS Z5000</i>	<i>;Bearbeitung ...</i>																		
<i>FASTABS X5000 Y5000</i>	<i>;Bearbeitung ...</i>																		
<i>COOLANT OFF</i>	<i>;Kühlmittelpumpe aus</i>																		
<i>SPINDLE OFF</i>	<i>;Spindel aus</i>																		
<i>GETTOOL 3</i>	<i>;Neues Werkzeug 3 holen</i>																		
<i>SPINDLE ON</i>	<i>;Spindel wieder einschalten</i>																		
<i>MOVEABS Z10000</i>	<i>;Sicherheitshöhe</i>																		
<i>FASTABS X150000 Y200000</i>	<i>;Positionieren ...</i>																		
<i>COOLANT ON</i>	<i>;Kühlmittelpumpe ein</i>																		
<p>Definition Bohrzyklus*)</p>	<p>DRILLDEF C<1> P<2> D<3> T<4> V<5> F<6> O<7> I<8> R<9> L<10> S<11></p> <p>Ein Bohrzyklus wird definiert, ohne die Bohrung unmittelbar auszuführen. Dieser Befehl wird benutzt, um alle Bohrparameter für den nachfolgenden Befehl DRILL zu setzen.</p> <p>Die Handhabung der Bohrparameter ist modal, d. h. wenn ein Parameter einmal gesetzt ist, bleibt diese Einstellung bis zum nächsten Aufruf des Befehls sowie Parameters gültig.</p>																		

Die Angabe der Bohrparameter kann in einer oder mehreren Zeilen erfolgen, z. B. um generelle Einstellungen zu Beginn der NC-Datei vorzunehmen und veränderliche Einstellungen vor Aufruf des jeweiligen Bohrbefehls.

Parameter	benutzt in Zyklus	Voreinstellwert	Bedeutung
C		1	Angabe des Bohrzyklus: 1 Einfaches Bohren 2 Tieflochbohren bzw. Ausräumen 3 Spanbrechen
P	1,2,3	0	Referenzebene, auf die sich alle weiteren Parameter des Bohrzyklus' beziehen. Die Referenzebene bezieht sich auf den definierten Werkstücknullpunkt, der Wert der Z-Koordinate gibt den Abstand dazu an. Die Angabe erfolgt in Mikrometern [µm]. Vorzeichen: Positiv, falls Z-Position oberhalb des Werkstücknullpunktes.
D	1,2,3	0	Tiefe des Bohrlochs in [mm], relativ zur Referenzebene. Vorzeichen: Positiv, wenn Zustellung nach unten.
T	1,2,3	0	Verzögerungszeit für Freischnitt bei Erreichen des Lochendes, Angabe in Millisekunden [ms].
V	1,2,3	1000	Vorschubgeschwindigkeit für Bohren, Angabe in [mm/s]. Als Geschwindigkeit für Eilbewegungen wird die zuletzt durch den Zwischenformat-Befehl FASTVEL eingestellte Eilgeschwindigkeit bzw. die Default-Eilgeschwindigkeit des Steuerungsprogramms benutzt.
F	2,3	0	Erste inkrementelle Zustelltiefe für Ausräumen und Spanbrechen, Angabe in [mm]. Vorzeichen: Positiv, wenn Zustellung nach unten.
O	2,3	0	Alle weiteren inkrementellen Zustelltiefen für Ausräumen und Spanbrechen, Angabe in [mm]. Vorzeichen: Positiv, wenn Zustellung nach unten.
I	2,3	0	Abnahme der inkrementellen Zustelltiefen pro Arbeitsgang. Vorzeichen: Positiv, wenn inkr. Zustelltiefe verringert wird.
R	3	0	Inkrementeller Rückzug für Spanbrechen oder Differenzhöhe für das Wiederanfahren der letzten Bohrtiefe im Eilgang bei Tieflochbohren. Angabe in [mm]. Vorzeichen: Positiv, da Rückzugwert angegeben wird.
L	1,2,3	0	Rückzugshöhe bezüglich der definierten Referenzebene nach dem Ende des Bohrens. Angabe in [mm]. Vorzeichen: Positiv, wenn Z-Position oberhalb des Werkstücknullpunktes.
S	1,2,3	0	Relative Sicherheitshöhe, die nach dem Ende des Bohrens relativ zur Referenzebene eingestellt wird. Angabe in [mm]. Vorzeichen: Positiv, wenn Z-Position oberhalb des Werkstücknullpunktes.

Beispiel siehe Befehl DRILL

Loch Bohren*)

DRILL X... Y...

Bohrung an der Position (X, Y).

Die Parameter des Bohrzyklus' und der auszuführende Bohrzyklus müssen zuvor mit dem Befehl DRILLDEF festgelegt worden sein.

Die Angabe der Bohrposition erfolgt mit den Parametern X und Y, die Einheit der Position ist Mikrometer [μm].

Beispiel

```

DRILLDEF P2000      ;Referenzebene 2 mm über dem definierten
                    ;Werkstücknullpunkt Z
DRILLDEF D20000     ;Bohrtiefe 20mm relativ zur Referenzebene
DRILLDEF T1000      ;Zeitverzögerung am Lochende 1 Sekunde
DRILLDEF V10000     ;Vorschubgeschw. für Bohren 10 mm/s
DRILLDEF F7000      ;Erste Zustelltiefe der Bohrung ist 7 mm
DRILLDEF O5000      ;Alle weiteren Zustelltiefen sind 5 mm und
DRILLDEF I1000      ;nehmen pro Zustellung um 1 mm ab
DRILLDEF R1000      ;Der inkrementelle Rückzug beträgt 1 mm
DRILLDEF L1000      ;Rückzugshöhe relativ zur Referenzebene
DRILLDEF S3000      ;Sicherheitshöhe relativ zur Referenzebene
DRILLDEF C1          ;Bohrzyklus 1, Einfaches Bohren, aktivieren
DRILL X10000 Y30000 ;An Position X = 10 mm, Y = 30 mm bohren
DRILLDEF C2          ;Bohrzyklus 2, Tieflochbohren, aktivieren
DRILL X30000 Y30000 ;An Position X = 30 mm, Y = 30 mm bohren
DRILLDEF C3          ;Bohrzyklus 3, Spanbrechen, aktivieren
DRILL X50000 Y30000 ;An Position X = 50 mm, Y = 30 mm bohren

```

Zylinderradius*)

CYL X(Y,Z,A)radius X(Y,Z,A)

Angabe des Radius' eines in einer Drehachse aufgespannten Zylinders
 Falls Sie eine Zwischenformatdatei für die Bearbeitung eines zylindrischen Drehkörpers gestalten wollen, können Sie diesen Befehl benutzen. Der Treiber rechnet dann die linearen Geschwindigkeiten auf der Zylinderoberfläche in entsprechende Drehgeschwindigkeiten um.

Der erste Parameter gibt die Maschinenachse an, die als Drehachse konfiguriert ist. Nachfolgend angegeben ist der Radius des eingespannten Zylinders, die Angabe erfolgt in Millimetern.

Der zweite Parameter gibt an, welche der drei Achsen X, Y oder Z eines kartesischen Koordinatensystems durch die Drehachse ersetzt wird.

Zu beachten: Auf einer Zylinderoberfläche kann keine Kreis- oder Helixinterpolation stattfinden.

In die Originalkonfiguration zurückschalten können Sie mit der Übergabe des Radius' 0, z. B.: CYL A0.

Beispiel

```

CYL A5000 A          ;Zylinderradius 5 mm, Drehachse ist Achse A

```

*) siehe Seite 14

Bearbeitung
anhalten*)

HALT

Anhalten der laufenden Bearbeitung und Umschalten in den Bearbeitungsmodus „Einzelschritt“.

Die Bearbeitung kann schrittweise oder kontinuierlich fortgesetzt werden, nachdem die an dieser Stelle notwendigen Aktionen von Benutzerseite ausgeführt wurden.

In der NC-Datei kann an entsprechender Stelle ein Kommentar stehen, der die erforderlichen Aktionen auf Seite des Benutzers beschreibt.

Beispiel

```

SPINDLE OFF           ;Spindel ausschalten
MOVEABS Z5000         ;Bearbeitung ...
FASTABS X5000 Y5000   ;Bearbeitung ...
                       ;Kommentar mit Anweisungen für den
                       ;Bediener, z. B.:
HALT                  ;Drehen Sie das Werkstücks um 90° nach
                       ;links und setzen Sie die Bearbeitung fort
MOVEABS Z10000        ;Sicherheitshöhe
FASTABS X15000 Y20000 ;Positionieren ...
SPINDLE ON TIME2000   ;Spindel wieder einschalten
  
```

Rampen*)

ACCEL X.... Y... Z.... A....

Einstellen der Beschleunigungswerte für die Achsen.

Die Angabe des Wertes erfolgt in Prozent. Der Wert der Beschleunigung kann in einem Bereich von 5 % ... 100 % liegen. Der maximale Beschleunigungswert (100 %) wurde in der Initialisierungsdatei eingestellt.

Spindel ein/aus*)

SPINDLE CW/CCW RPMxxx RPSxxx TIMExxx

Ein-/Ausschalten der Antriebsspindel. CW bzw. CCW bestimmen die Drehrichtung der Spindel, der Parameter RPM (RPS) die Drehzahl in Umdrehungen/Minute [U/s].

Eine Programmverzögerung für den Anlauf der Spindel auf Nenn-drehzahl kann durch den Parameter TIME angegeben werden.

Die Angabe der Zeit erfolgt in Millisekunden [ms].

Kühlmittel ein/aus*)

COOLANT ON/OFF

Ein-/Ausschalten der Kühlmittelpumpe.

Die Festlegung der Art der Kühlmittelpumpensteuerung erfolgt im Interpreterteil des NC-Steuerungsprogramms.

Def. Werkzeug*)

TOOLDEF Tx Ly Rz

Bekanntgabe der verwendeten bzw. bestückten Werkzeuge.
(Für den späteren Gebrauch)

*) siehe Seite 14

Init WZWechsel*) **INITTOOL**
Initialisieren des angeschlossenen Werkzeugwechslers/des Werkzeug-
wechselteils der CNC-Software. Durchführen von WZ-Wechsel-Initiali-
sierung, Feststellen der aktuellen Konfiguration, Referenzläufe etc.
(Für den späteren Gebrauch)

Programmende **PROGEND**der NC-Datei
Markiert das Ende der NC-Datei. Abhängig von den Einstellungen im
Steuerungsprogramm wartet der Interpreter entweder auf Anweisun-
gen zum Neustart der Bearbeitung oder das Ausgabefenster wird
automatisch geschlossen und die Haupteingabemaske des
Steuerungsprogramms wieder angezeigt.

Beispiel

<i>SPINDLE OFF</i>	<i>;Spindel ausschalten</i>
<i>COOLANT OFF</i>	<i>;Kühlmittelpumpe ausschalten</i>
<i>REF XYZ</i>	<i>;Achsen an definierte Positionen fahren</i>
<i>PROGEND</i>	<i>;Programmende kennzeichnen</i>

*) Diese Funktion wird von I5EIN nicht unterstützt, da I5EIN nur ein Einstell- und Testprogramm ist.

**Beispiel für eine NCP-Quelldatei**

Die Beispieldatei wurde mit einem HP/GL-Konverter erstellt und zeigt das Ausfräsen von drei Rechtecken.

```
N000001 IMF_PBL_V1.0 - HPREMOTE V1.32 - PP FILE
N000002 ;*****
N000003 ; 3RECTS.NCP Fri Mar 01 12:04:29 1996
N000004 ;*****
N000005 VEL 5000
N000006 FASTVEL 50000
N000007 MOVEABS Z-3000
N000008 VEL 8000
N000009 FASTVEL 50000
N000010 FASTABS X53375 Y0 Z-3000
N000011 MOVEABS Z5000
N000012 VEL 12000
N000013 MOVEREL Y69625
N000014 MOVEREL X67625
N000015 MOVEREL Y-69625
N000016 MOVEREL X-67625
N000017 VEL 10000
N000018 MOVEABS Z-3000
N000019 VEL 5000
N000020 FASTVEL 50000
N000021 FASTABS X30625 Y54000 Z-3000
N000022 MOVEABS Z7000
N000023 VEL 12000
N000024 MOVEREL Y76250
N000025 MOVEREL X72875
N000026 MOVEREL Y-76250
N000027 MOVEREL X-72875
N000028 VEL 10000
N000029 MOVEABS Z-3000
N000030 VEL 5000
N000031 FASTVEL 50000
N000032 FASTABS X0 Y16875 Z-3000
N000033 MOVEABS Z1000
N000034 MOVEREL Y65875
N000035 MOVEREL X61250
N000036 MOVEREL Y-65875
N000037 MOVEREL X-61250
N000038 MOVEABS Z-3000
N000039 PROGEND
```